Solution to Exercise 3.2.9

Let $G := \mathrm{J}_2$. We proceed similarly as in Example 3.2.25 and compute (using the GAP-command `InducedCyclic`)

$$\texttt{ind} := \bigcup_{g \in G} \{\, \lambda^G \mid \lambda \in \operatorname{Irr}(\langle g \rangle) \,\} :$$

```
gap> t := CharacterTable("J2");;
gap> irr := [ List( [1..21] , x ->1) ];; ind := InducedCyclic( t, "all");;
gap> red := Reduced( t, irr, ind );;  l := LLL( t, red.remainders );;
gap> Append( irr, l.irreducibles );  List( irr, y -> y[1] );
[ 1, 336 ]
```

By reducing `ind` with the trivial character and using the LLL-algorithm we have already found an irreducible character of G of degree 336.

```
gap> r := l.remainders;; M := MatScalarProducts( t, r, r );;
gap> oes := OrthogonalEmbeddingsSpecialDimension( t, r, M, 19 );;
gap> Append( irr, oes.irreducibles );  List( irr, y -> y[1] );
[ 1, 336, 36, 63, 70, 70, 90, 126, 175, 225, 300 ]
```

The command `OrthogonalEmbeddingsSpecialDimension` (see Example 2.8.16) yieded another nine irreducible characters of G. We can obtain two more irreducibles using symmetrizations of the irreducible characters obtained so far:

```
gap> r := oes.remainders;;
gap> for i in [2,3,4] do
>       Append( r, Symmetrizations( t,irr{[2..11]}, i ) );
>       r := Reduced( t, irr, r).remainders;
>    od;
gap> l:=LLL( t, r );; Append( irr, l.irreducibles ); List( irr, y -> y[1] );
[ 1, 336, 36, 63, 70, 70, 90, 126, 175, 225, 300, 160, 288 ]
```

Now only eight irreducibles are missing. We use `OrthogonalEmbeddings` again:

```
gap> r := l.remainders;; M := MatScalarProducts( t, r, r );;
gap> oe := OrthogonalEmbeddings( M , 8 );;
gap> Length(oe.solutions);
2
gap> Xli:= List( oe.solutions, x ->oe.vectors{x} );;
>    irrcands :=  List (Xli, x -> (TransposedMat(x) )^-1 * r );;
>    for i in [1,2] do
>      for y in irrcands[i] do
>        if y[1] < 0 then y := -y; fi;
>      od;
>      SortParallel( List(irrcands[i], y -> y[1]) , irrcands[i] );
>    od;
gap> ff := Filtered([1..21] , i -> ForAny([1..8] , j -> irrcands[1][j][i]
>        <>  irrcands[2][j][i]) );   ClassNames(t){ff};
[ 7, 8, 9, 10, 15, 16 ]
[ "5a", "5b", "5c", "5d", "10a", "10b" ]
```

We have obtained two candidates for the list of the missing irreducible characters of G. They differ only on six conjugacy classes (of elements of order 5 and 10). It turns out that the characters in `irrcands[1]` coincide with those printed in the Atlas, whereas those in `irrcands[2]` differ by a permutation of the classes:

```
gap> TransformingPermutations(irrcands[1],irrcands[2]);
rec( columns := (9,10)(17,18), rows := (1,2)(5,6),
  group := Group([ (5,12), (7,8)(9,10)(15,16)(17,18)(20,21) ]) )
gap> pc := (9,10)(17,18);; pr := (1,2)(5,6);;
gap> ForAll([1..8] , j -> ForAll([1..21], i -> irrcands[1][j][i] =
>                                           irrcands[2][j^pr][i^pc] ) );
true
```